EXAMPLE 2 - Data Coverage and Vintage:

Specialist's Request

I want a map showing the kind of coverage - {SITECOND} only, or "full" ({SITECOND} + {COMPOSIT} + {STRUCTUR}) - and the year of the data.

We should have a page-size photo ready map for handouts, and a wall map for discussion

We only need to use two tables for this example as any [stand]s that are in {COMPOSIT} are also in {STRUCTUR}.

There are several ways to skin this cat, so how you do it is up to you.

-Any [stand]s that occur in the table {STRUCTUR} have "Full" coverage.

-Any [stand]s that occur in the table {SITECOND} but do not occur in the table {STRUCTUR} have "SITECOND Only" coverage.

-The data vintage is identified by the year which is the first two characters in the [date] field in both the {SITECOND} and the {STRUCTUR} tables.

For the legend, I would like a different color for each year with the shading hatched for "SITECOND Only" and solid for "Full" coverage. Please do include [stand]s where the [date] field is blank and either leave the year blank in the legend, or make the year "Unknown".

Also, plot something on the maps to indicate location - such as District boundaries or major highways.

EXAMPLE 2 - Solution:

To start this exercise, it may be useful to determine which years can be found in the [date] field of the {SITECOND} table. Use this table because it contains all stands, unlike the {STRUCTUR} table. The easiest way to find these unique values is to use Informix directly. Be sure to set your terminal type as discussed in the introduction:

$ *export TERM=vt100*
$ *isql*

An introductory screen will appear, and then the following menu:

INFORMIX-SQL: *Form* Report Query-Language User-menu Database Table Exit
Run, Modify, Create, or Drop a form.

---------- Press CTRL-W for Help --------

Type 'D' for Database, or use the arrow keys to highlight 'Database' and hit 'Enter'. Then:

DATABASE: *Select* Create Drop Exit
Select a database to work with.

---------- oaesis@or3020_01 ------- Press CTRL-W for Help --------

Type 'S' or highlight 'Select' and press 'Enter' to start the database selection menu:

SELECT DATABASE >>
Select a database with the Arrow Keys, or enter a name, then press Return.

---------- oaesis@or3020_01 ------- Press CTRL-W for Help --------

basetrack@or3020_01

idatrack@or3020_01

oaesis@or3020_01

pin@or3020_01

Highlight 'oaesis@or3020_01' and press 'Enter' to select the OAESIS database. Exit the 'Database' menu with 'E' or by highlighting 'Exit'. Back at the initial menu, select 'Query-Language' to bring up the SQL menu:

SQL: New Run Modify Use-editor Output Choose Save Info Drop Exit
Enter new SQL statements using the SQL editor.

---------- oaesis@or3020_01 ------- Press CTRL-W for Help --------

Now is the time to enter the SQL statement to find the various years in the database. Pick either 'New' or 'Use-editor'. 'New' and later, 'Modify' will show an Informix editing screen. The 'Use-editor' choice will allow you to use an editor such as 'vi' to enter new SQL queries and edit them. Enter the following text:

select distinct date[1,2] from sitecond;

Exit the editor and select 'Run'. This executes the query. If you made a typing error or want to change the query, use your editor of choice to make the changes and 'Run' the query again. Look at the range of dates that you must look for when making the map.

When done with Informix, use 'Exit' as necessary to return to the UNIX prompt. Reset your terminal setting:

$ *export TERM=aixterm*

Now we can begin the Arc/Info session. Notice that this request uses two tables - {SITECOND} and {STRUCTUR}. As before, relates to the Informix tables must be established in Arc/Info. We established a relate to {SITECOND} in the previous example, but that relate is lost when you leave Arc/Info. Below, we will establish the relates and save them to a file so we can call them back at another time, or, in an AML:

Arc: *connect informix oaesis*
Connection to INFORMIX successful.
Arc: *relate add*
Relation Name: *sitecond.rel*

Table Identifier: ***sitecond***
Database Name: ***informix***
INFO Item: ***stand***
Relate Column: ***stand***
Relate Type: ***linear***
Relate Access: ***ro***
Relation Name: ***structur.rel***
Table Identifier: ***structur***
Database Name: ***informix***
INFO Item: ***stand***
Relate Column: ***stand***
Relate Type: ***linear***
Relate Access: ***ro***
Relation Name: *<cr>*
Arc: ***relate save ex2_rel.save***
2 Relates saved to file ex2_rel.save

Run the AML to create the map. This AML is similar to the last program. Review the comments throughout the program to see how it works:

Arc: ***&r example2.aml***

The output is an Arc graphics, or .gra, file called **example2.gra**.

Example 2 - Data Coverage & Vintage



Data Coverage & Vintage

Note: Hatched Shading Denotes SITECOND Only
Solid Shading Is Full Database Coverage

```
/* EXAMPLE2.AML
/*
/* Steve Salas - GDAD/Pacer-Infotec/OSO/BLM
/*
/* November 18, 1996
/*

/* SETUP STUFF, CHECK CONNECTION TO INFORMIX
&if [show connects]x nc informix &then~
 connect informix oaesis
relate restore ex2_rel.save
&s old_apath := [show &amlpath]
&amlpath /or_tools/pdtk_ap/aml /or_tools/pdtk_ap/aml_local

/* ENTER ARCPLOT
display 0
arcplot
display 1040
example2.gra

/* FIND THE EXTENT OF THE GRID AND SET MAPEXTENT THIS WAY.  THIS
/* WILL ENSURE THE GRID GOES ALL THE WAY TO THE MAPLIMITS.
&describe ../example1/pvt_merge
&s xmin := %grd$xmin%
&s ymin := %grd$ymin%
&s xmax := %grd$xmax%
&s ymax := %grd$ymax%

mape %xmin% %ymin% %xmax% %ymax%

/* RUN THE PLOT FRAME GENERATOR
&r start_blm_plot 1000000~
 [quote Example 2 - Data Coverage & Vintage]~
 doiblm_logo utm_bar11.aml none 'S. Salas'

/* USE SOME REVERSE LOGIC HERE... FIND STANDS THAT ARE IN STRUCTUR,
/* FLIP THE SET TO FIND THINGS THAT ARE NOT IN STRUCTURE, SO THESE ARE
/* ONLY IN SITECOND.
reselect ../oaesis poly stand = structur.rel//stand
writeselect structur.wrs ../oaesis poly  /* POLYS FOUND IN STRUCTUR = FULL
nsel ../oaesis poly                      /* REMAINING POLYS ARE...
writeselect sitecond.wrs ../oaesis poly  /* POLYS FOUND IN SITECOND ONLY!!!

/* SITECOND ONLY POLYGONS - HATCH
shadeset blmoso1.shd
readselect sitecond.wrs
&s year := 76
&do sym &list 01 04 17 86 80 82 74 08 62 60 49 42 44 43 21 30 27 35 11 14
  &type YEAR 19%year% - SITECOND ONLY
  reselect ../oaesis poly sitecond.rel//date lk [quote %year%*]
  polygonshades ../oaesis %sym%
  shadesym 301
  &if %sym% = 01 &then shadecolor white
  &else shadecolor black
  shadeseparation .025 /* ALTER THE SPACING
  shadeput 301
  polygonshades ../oaesis 301
  clearsel ../oaesis poly
  readselect sitecond.wrs
  &s year := %year% + 1
```

```
 &end
&type BLANK YEAR IN SITECOND TABLE
resel ../oaesis poly sitecond.rel//date = ''
shadesym 7
polygonshades ../oaesis 7
shadesym 301
shadeseparation .025
shadecolor black
shadeput 301
polygonshades ../oaesis 301
clearselect ../oaesis poly

/* FULL COVERAGE - SOLID
&r mkleg i  /* INITIALIZE LEGEND
shadeset blmoso1.shd
readselect structur.wrs
&s year := 76
&do sym &list 01 04 17 86 80 82 74 08 62 60 49 42 44 43 21 30 27 35 11 14
  &type YEAR 19%year% - FULL
  reselect ../oaesis poly structur.rel//date lk [quote %year%*]
  shadesym %sym%
  polygonshades ../oaesis %sym%
  &r mkleg p [quote 19%year%]  /* ADD TO LEGEND
  clearsel ../oaesis poly
  readselect structur.wrs
  &s year := %year% + 1
 &end
/* NEXT LINES NOT DONE, INFORMIX QUERY REVEALS NO EMPTY DATES IN STRUCTUR
/* reselect ../oaesis poly structur.rel//date = ''
/* asel ../oaesis poly stand ne structur.rel//stand and~
/*  stand ne sitecond.rel//stand
/* polygonshades ../oaesis 7
shadesym 7
&r mkleg p 'Unknown'
clearselect ../oaesis poly

/* SKIP THE LEGEND OVER THE BARSCALE
&r mkleg c
&r mkleg c
&r mkleg c
&r mkleg c

/* SET THE LIBRARY FOR OTHER DATA, USE PROPER MAP PROJECTION
library q100
mapprojection /or_tools/arc/projections/lam120.5utm11.prj

/* PLOT ROADS.  PUT IN LEGEND.
lineset blmoso1.lin
/* &goto skip_road
resel .roads arc ushwy ne '' or sthwy ne ''
linesym 102
arcs .roads
&r mkleg l 'US/State Highway'
clearsel .roads arc

resel .roads arc inthwy ne ''
linesym 302
linesize .04
arcs .roads
&r mkleg l
```

```
&r mkleg r
linesym 1
linetype wide
linesize .045
linehollow .045
arcs .roads
&r mkleg 1
&r mkleg r
linesym 1
arcs .roads
&r mkleg 1 'Interstate Highway'
&label skip_road

/* DONE USING Q100, BACK TO UTM11 PROJECTION.
mapprojection off

/* PLOT DISTRICT BOUNDARIES
linesym 310
polys ../orwadob11
&r mkleg 1
&r mkleg r
linesym 714
polys ../orwadob11
&r mkleg 1 'District Boundary'

/* PLOT THE NEATLINE
linesym 1
box [show maplimits]

/* MAKE A LAT/LONG GRID.
textsym 22
textsize .1
mapprojection off
&r mkgeogrid1 1 mapextent /or_tools/arc/projections/geoutm11.prj~
 1 0 1 0 -1 +1 1 0 1 0 -3 +3

/* NORTH ARROW
plot /or_tools/arc/ap_elements/arrows_sym/arrow8.gra box 22.6 .8 23.0 1.8

/* MORE LEGEND INFO
textsym 24
textsize .08
move 24.5 1.6
text 'Note: Hatched Shading Denotes SITECOND Only'
move 24.5 1.52
text '           Solid Shading Is Full Database Coverage'

/* LEAVE AP, ENDS PLOTFILE
quit

/* RESTORE USER'S AMLPATH
&amlpath %old_apath%

&return
```